Solved In Perl 6

Jonathan Worthington

YAPC::Asia 2009

Solved in Perl 6

What is this talk about?

- How to do a range of everyday tasks using Perl 6
- A chance to show off new Perl 6 features
- All examples shown today work in Rakudo

このトークの内容

- よく使う処理をPerl 6で書いてみよう
- Perl 6の新しい機能も紹介するよ
- サンプルはすべてRakudoで動作確認済

Solved in Perl 6

Problem

Say "Hello, world"
「Hello, world」を出力する

Solution

```
say "Hello, world!"
```

Output

```
Hello, world!
```

Solved in Perl 6

Problem

Read input from the console
コンソールからの読み込み

Solution

```
print "Enter your name: ";
my $name = $*IN.get;
say "こんにちは $name!";
```

Output

```
Enter your name: Jonathan
こんにちは Jonathan!
```

Solved in Perl 6

Problem

Check a value is in a given range
値が指定された範囲内にあるか確認

Solution 1

```
print "Enter a number between 1 and 10: ";
my $number = $*IN.get;
unless 1 <= $number <= 10 { say "Oh no!" }
```

Output

```
Enter a number between 1 and 10: 3
Enter a number between 1 and 10: 42
Oh no!
```

Solved in Perl 6

Problem

Add up a list of numbers
値のリストの合計を求める

Solution

```
my @nums = 1, 5, 7, -2, 3, 9, 11, -6, 14;
say [+] @nums;
```

Output

```
42
```

Solved in Perl 6

Problem

Check if a list is sorted
リストがソートされているか確認

Solution

```
my @a = 1, 1, 2, 3, 5, 8;
my @b = 9, 4, 1, 16, 36, 25;
if [<=] @a { say '@a is sorted' }
if [<=] @b { say '@b is sorted' }
```

Output

```
@a is sorted
```

Solved in Perl 6

Problem

Get a Perl-ish representation of a data structure
データ構造をPerl風の表記に

Solution

```
my @a = 1, 2, 3;
push @a, { x => 42, y => 100 };
say @a.perl;
```

Output

```
[1, 2, 3, {"y" => 100, "x" => 42}]
```

Solved in Perl 6

Problem

Iterate over a list
リストのイテレーション

Solution

```
my @cities = <Tokyo Kyoto Nara>;
for @cities -> $city {
    say "I'll visit $city";
}
```

Output

```
I'll visit Tokyo
I'll visit Kyoto
I'll visit Nara
```

Solved in Perl 6

Problem

Iterate over the keys and values of a hash
ハッシュのキー/値のイテレーション

Solution

```
my %distances = Kyoto => 514, Nara => 850;
for %distances.kv -> $city, $distance {
    say "$city is $distance km away";
}
```

Output

```
Nara is 850 km away
Kyoto is 514 km away
```

Solved in Perl 6

Problem

Check if any of a list of test scores is a pass
テスト結果のリストに合格が含まれているか確認

Solution

```
my @a = 75, 47, 90, 22, 80;
my @b = 61, 77, 94, 82, 60;
my @c = 45, 59, 33, 11, 19;
if any(@a) >= 60 { say "Some passes in A" }
if any(@b) >= 60 { say "Some passes in B" }
if any(@c) >= 60 { say "Some passes in C" }
```

Output

```
Some passes in A
Some passes in B
```

Solved in Perl 6

Problem

Check if all of a list of test scores are passes
テスト結果のリストがすべて合格か確認

Solution

```
my @a = 75, 47, 90, 22, 80;
my @b = 61, 77, 94, 82, 60;
my @c = 45, 59, 33, 11, 19;
if all(@a) >= 60 { say "All passes in A" }
if all(@b) >= 60 { say "All passes in B" }
if all(@c) >= 60 { say "All passes in C" }
```

Output

```
All passes in B
```

Solved in Perl 6

Problem

Check if none of a list of test scores is a pass
テスト結果のリストに合格がないことを確認

Solution

```
my @a = 75, 47, 90, 22, 80;
my @b = 61, 77, 94, 82, 60;
my @c = 45, 59, 33, 11, 19;
if none(@a) >= 60 { say "No passes in A" }
if none(@b) >= 60 { say "No passes in B" }
if none(@c) >= 60 { say "No passes in C" }
```

Output

```
No passes in C
```

Solved in Perl 6

Problem

Get a random item from a list
リストからランダムに抽出

Solution

```
my @drinks = <sake beer vodka>;
say "Tonight I'll drink { @drinks.pick }";
```

Output (results should vary ;-))

```
Tonight I'll drink vodka
```

Solved in Perl 6

Problem

Shuffle a list into a random order
リストをランダムな順序にシャッフル

Solution

```
my @competitors = <Tina Lena Owen Peter>;
my @order = @competitors.pick(*);
@order>>.say;
```

Output (results should vary ;-))

```
Peter
Lena
Owen
Tina
```

Solved in Perl 6

Problem

Write and call a subroutine with parameters
パラメータ付きサブルーチンの書き方と呼び方

Solution

```
sub greet($name) {
    say "こんにちは $name!";
}
greet("Patrick");
```

Output

```
こんにちは Patrick!
```

Solved in Perl 6

Problem

Write a subroutine taking an array and a hash
配列とハッシュを取るサブルーチンの書き方

Solution

```
sub example(@a, %h) {
    say @a.elems;
    say %h.keys;
}
my @nums = 42, 57, 74;
my %mapping = a => 1, b => 2;
example(@nums, %mapping);
```

Output

```
3
ab
```

Solved in Perl 6

Problem

Write a subroutine that only takes a number
数値のみ取るサブルーチンの書き方

Solution

```
sub double(Num $n) { 2 * $n }
say double(21);
say double("oh no I'm not a number");
```

Output

```
42
Parameter type check failed; expected Num,
but got Str for $n in call to double
```

Solved in Perl 6

Problem

Use multi-subs to react differently by type
型によって動作が異なる多重サブルーチンの書き方

Solution

```
multi double(Num $n) { 2 * $n }
multi double(Str $s) { $s x 2 }
say double(21);
say double("boo");
```

Output

```
42
booboo
```

Solved in Perl 6

Problem

Compute factorial (recursively)
階乗を計算する（再帰で）

Solution

```
multi fact($n) { $n * fact($n - 1) }
multi fact(0)  { 1 }
say fact(1);
say fact(10);
```

Output

```
1
3628800
```

Solved in Perl 6

Problem

Compute factorial (using a meta-operator)
階乗を計算する（メタ演算子で）

Solution

```
sub fact($n) { [*] 1..$n }
say fact(1);
say fact(10);
```

Output

```
1
3628800
```

Solved in Perl 6

Problem

Add a new factorial operator (so 10! works)

新しい階乗演算子を追加する(10!と書けるように)

Solution

```
sub postfix:<!>($n) { [*] 1..$n }
say 1!;
say 10!;
```

Output

```
1
3628800
```

Solved in Perl 6

Problem

Declare a class with attributes and a method
アトリビュートとメソッドを持つクラスの宣言

Solution

```
class Product {
    has $.name;   # Attr + accessor
    has $!price;  # Attr only
    has $.discount is rw;
                  # Attr + lvalue accessor
    method get_price {
        return $!price - $!discount;
    }
}
```

Problem

Instantiate a class and call a method on it
クラスのインスタンス化とメソッドの呼び出し

Solution

```
my $prod = Product.new(
    name        => "Beer",
    price       => 500,
    discount    => 60
);
say $prod.get_price;
```

Output

```
440
```

Solved in Perl 6

Problem

Get/set attributes through accessors
アクセサを使ったアトリビュートの取得/設定

Solution

```
say $prod.name;
$prod.discount = 40;
say $prod.get_price;
$prod.name = 'Wine';
```

Output

```
Beer
460
Cannot assign to readonly variable.
```

Solved in Perl 6

Problem

Call a method on every object in a list
リストのすべてのアイテムにメソッド呼び出し

Solution

```
my @products =
  Product.new(name => 'Beer', price => 500),
  Product.new(name => 'Wine', price => 450),
  Product.new(name => 'Vodka', price => 1600);
@products>>.name>>.say;
```

Output

```
Beer
Wine
Vodka
```

Solved in Perl 6

Problem

Introspect a class to find its methods
クラスの中身を覗いてメソッドを探す

Solution

```
my @meths = Product.^methods(:local);
@meths>>.name>>.say;
```

Output

```
get_price
discount
name
```

Solved in Perl 6

Problem

Sort an array of objects by result of a method

メソッドの結果を利用してオブジェクトの配列をソートする

Solution (Example 1)

```
my @products =
  Product.new(name => 'Beer', price => 500),
  Product.new(name => 'Wine', price => 450),
  Product.new(name => 'Vodka', price => 1600);
@products.sort(*.name)>>.name>>.say;
```

Output (Example 1)

```
Beer
Vodka
Wine
```

Solved in Perl 6

Problem

Sort an array of objects by result of a method

メソッドの結果を利用してオブジェクトの配列をソートする

Solution (Example 2)

```
my @products =
  Product.new(name => 'Beer', price => 500),
  Product.new(name => 'Wine', price => 450),
  Product.new(name => 'Vodka', price => 1600);
@products.sort(*.get_price)>>.name>>.say;
```

Output (Example 2)

```
Wine
Beer
Vodka
```

Solved in Perl 6

Problem

Find minimum and maximum values from a list
リストから最小値、最大値を見つける

Solution (Example 1)

```
my @temperatures = -3, 5, 7, 2, -1, -4, 0;
say "Minimum was " ~ @temperatures.min;
say "Maximum was " ~ @temperatures.max;
```

Output (Example 1)

```
Minimum was -4
Maximum was 7
```

Problem

Find minimum and maximum values from a list
リストから最小値、最大値を見つける

Solution (Example 2)

```
my @products =
  Product.new(name => 'Beer', price => 500),
  Product.new(name => 'Wine', price => 450),
  Product.new(name => 'Vodka', price => 1600);
say "Cheapest: " ~ @products.min(*.get_price).name;
say "Costliest: " ~ @products.max(*.get_price).name;
```

Output (Example 2)

```
Cheapest: Wine
Costliest: Vodka
```

Solved in Perl 6

Problem

Paper, Scissor, Stone game
じゃんけん

Solution (Part 1)

```
class Paper { }
class Scissor { }
class Stone { }
multi win(Paper,   Stone)   { "Win" }
multi win(Scissor, Paper)   { "Win" }
multi win(Stone,   Scissor) { "Win" }
multi win(::T,     T)       { "Draw" }
multi win(Any,     Any)     { "Lose" }
```

Problem

Paper, Scissor, Stone game
じゃんけん

Solution (Part 2)

```
say win(Paper, Paper);
say win(Scissor, Stone);
say win(Stone, Scissor);
```

Output

```
Draw
Lose
Win
```

Solved in Perl 6

Want to play with Perl 6?

- Get Rakudo:
  http://www.rakudo.org/
- Lots of Perl 6 resources:
  http://www.perl6.org/

Perl 6を使ってみたくなったら

- Rakudoをダウンロード:
  http://www.rakudo.org/
- さまざまなPerl 6関連リソース:
  http://www.perl6.org/

ありがとう
ございます